

```

*****
/* Date:      4/4/95  3:40 pm
/*
/* Macro:     UTLNGCHZ.WCM
/*
/* Purpose:   Provide a language menu dialog box that allows
/*            the user to switch "Gamma Productions" UniType
/*            program language without leaving WordPerfect.
/*            In addition to switching the UniType language the
/*            font at the cursor insertion point is switched to the
/*            ILI configured UI font for the new language.
*****
/* WordPerfect's standard behavior is such that when you change fonts at
/* a cursor insertion point, you are actually changing the font from the cursor
/* point all the way to the next font change in the text. To override this
/* behavior, this macro records the existing font at the cursor point before
/* making the font change. This font is stored in the variable: oldFont. After
/* making the font change to the new font a dummy character and a font
/* change back to the oldFont are inserted. Then the cursor is moved back
/* before the dummy character and the dummy character is deleted.
/*
/* This KLUDGE effective allows the macro to change the font of only
/* new characters that will be typed after evoking the macro.
/*
/* If you want the standard WordPerfect behavior you would edit the
/* second code line of this macro to say: WordPerfectKludge:=False
/*
/* Note:      If the macro is assigned to a keystroke, pressing that keystroke
/*            will NOT run the macro when UniType is active and the UniType
/*            language is in "Word" or "Phrase" mode and there is text pending
/*            in the UniType edit buffer.
*****
/* Application (A1; "WordPerfect"; Default; "US")
Display(State: Off!)
WordPerfectKludge:=True
WM_COMMAND:= 273
WM_SYSCOMMAND:= 274
LBN_DBLCLK:= 2
langList = "Selection"
langStr = 0
DDEInitiate(hChannel; "UniType"; "UniType")
If(hChannel != 0)
DialogDefine("Dlg"; 50; 50; 108; 138; Percent! | OK! | Cancel!; "UniType Language")
DialogAddText("Dlg"; 100; 6; 4; 129; 13; Left!; "Language:")
DialogAddListBox("Dlg"; 101; 6; 16; 90; 84; Sorted!; langList)
// fill the listbox with lang names

```

```

langStr = DDERequest(hChannel; "ut_get_language 0")
For(i;1; langStr <> ""; i+1)
  DialogAddListItem("Dlg"; 101; langStr)
  langStr = DDERequest(hChannel; "ut_get_language " + i)
EndFor
utactive = DDERequest(hChannel; "ut_get_active")
// disable UniType while during language menu
If(utactive = "1")
  DDEExecute(hChannel; "ut_set_active 0")
EndIf
DialogHandle(hWndList; "Dlg"; 101)
DialogDisplay("Dlg"; 101; DlgProc)
// Select first listbox item
DLLLoad(UserLink; "User")
DLLCall(UserLink; "SendMessage"; nVoid:INTEGER; {LoWord(hWndList); LoWord(1031);
LoWord( 0); 0})
DllFree(UserLink)
// wait for user to end the dialog
bDone:= False
While(bDone = False)
EndWhile
DialogDestroy("Dlg")
// re-enable UniType if it was disabled
If(utactive = "1")
  DDEExecute(hChannel; "ut_set_active 1")
EndIf
// listbox selection should be in: langStr
If(langStr <> "")
  DDEExecute(hChannel; "ut_set_language " + langStr)
  uifontStr = DDERequest(hChannel; "ut_get_utfont " + langStr)
  SelectOff()
  CodesRevealed:= ?RevealCodesActive
  If(CodesRevealed = True)
    RevealCodes(Off!)
  EndIf
  oldfontStr = ?Font
  Font(uifontStr)
  //////////////////////////////////////
  // see comment at top of Macro
  If(WordPerfectKludge)
    Type("?")
    Font(oldfontStr)
    PosCharPrevious()
    DeleteCharNext()
  EndIf
  //////////////////////////////////////

```

```

If(CodesRevealed = True)
  RevealCodes(On!)
EndIf
EndIf
DDETERMINATE(hChannel)
Else
  MessageBox(mbResult; "Warning!"; "Unable to establish DDE connection with UniType." +
  NTOC(0F90Ah) + NTOC(0F90Ah) + "Please run the UniType program and try
  again."; IconExclamation!)
Endif
Quit
////////////////////////////////////
// Dialog Procedure
LABEL(DlgProc)
If(DlgProc[3] = "OKBbtn")
  DialogUndisplay("Dlg"; 101)
  langStr = langList
  bDone = True
  Return
EndIf
If(DlgProc[3] = "CancelBbtn")
  bDone = True
  Return
EndIf
Switch(DlgProc[5])
  CaseOf WM_COMMAND :
    Switch(LOWRD(DlgProc[7]))
      CaseOf hWndList :
        Switch(HIWRD(DlgProc[7]))
          CaseOf LBN_DBLCLK :
            DialogUndisplay("Dlg"; 101)
            langStr = langList
            bDone = True
          EndSwitch
        EndSwitch
      CaseOf WM_SYSCOMMAND :
        bDone = True
    EndSwitch
  Return

Function LOWRD(x)
  x = x & 65535
  return(x)
EndFunc

Function HIWRD(x)

```

```
x = x>>16  
return(x)  
EndFunc
```